

Partial Differential Equations with MAPLE

By

Gilberto E. Urroz, Ph.D., P.E.

Distributed by

 *infoClearinghouse.com*

©2001 Gilberto E. Urroz
All Rights Reserved

Download at InfoClearinghouse.com

15 Partial Differential Equations

This chapter introduces basic concepts and definitions for partial differential equations (PDEs) and solutions to a variety of PDEs. Applications of the method of separation of variables are presented for the solution of second-order PDEs. The application of this method involves the use of Fourier series that were introduced in Chapter 13. This chapter also introduces Fourier transforms which can be used to simplify the solution of some PDEs. The aim of the chapter is to demonstrate Maple applications in the solution of well-known PDEs from mathematical physics, rather than being a comprehensive presentation on partial differential equations.

Definitions

■ Equations involving one or more partial derivatives of a function of two or more independent variables are called partial differential equations (PDEs).

■ Well known examples of PDEs are the following equations of mathematical physics in which the notation: $u = \partial u / \partial x$, $u_{xy} = \partial u / \partial y \partial x$, $u_{xx} = \partial^2 u / \partial x^2$, etc., is used:

[1] One-dimensional wave equation: $u_{tt} = c^2 u_{xx}$

[2] One-dimensional heat equation: $u_t = c^2 u_{xx}$

[3] Laplace equation: $u_{xx} + u_{yy} = 0$, (2-D), or $u_{xx} + u_{yy} + u_{zz} = 0$ (3-D)

[4] Poisson equation: $u_{xx} + u_{yy} = f(x, y)$, (2-D), or $u_{xx} + u_{yy} + u_{zz} = f(x, y, z)$ (3-D)

■ The order of the highest derivative is the order of the equation. For example, all of the PDEs in the examples shown above are of the second order.

■ A PDE is linear if the dependent variable and its functions are all of first order. All of the PDEs shown above are also linear.

■ A PDE is homogeneous if each term in the equation contains either the dependent variable or one of its derivatives. Otherwise, the equation is said to be non-homogeneous. Equations [1], [2], and [3] above are homogeneous equations. Equation [4] is non-homogeneous.

■ A solution of a PDE in some region R of the space of independent variables is a function, which has all the derivatives that appear on the equation, and satisfies the equation everywhere in R. For example, $u = x^2 - y^2$, $u = e^x \cos(y)$, and $u = \ln(x^2 + y^2)$, are all solutions to the two-dimensional Laplace equation (equation [3] above).

Example 1 -- Verify that $u = \sin wct \sin wx$ satisfies the wave equation $u_{tt} = c^2 u_{xx}$.

Using Maple:

```
> restart:PDEwave:=diff(u(x,t),t$2) = c^2*diff(u(x,t),x$2);
> u:=(x,t) -> sin(w*c*t) *sin (w*x);
> PDEwave;
```

$$-\sin(wct) w^2 c^2 \sin(wx) = -\sin(wct) w^2 c^2 \sin(wx)$$

Example 2 -- Determine the value of c so that the function $u = e^{-t} \sin 3x$ satisfies the heat equation.

Using Maple:

```
> restart:HeatEqn:=diff(u(x,t),t) = c^2*diff(u(x,t),x$2);
> u:=(x,t) -> exp(-t)* sin(3*x);
> HeatEqn;
```

$$-e^{(-t)} \sin(3x) = -9c^2 e^{(-t)} \sin(3x)$$

```
> solve(HeatEqn,{c});
```

$$\left\{c = \frac{1}{3}\right\}, \left\{c = -\frac{1}{3}\right\}$$

Example 3 -- Verify that the function $u = \tan^{-1}(y/x)$ is a solution of Laplace's equation in two dimensions.

```
> restart:LaplaceEqn:=diff(u(x,y),x$2)+diff(u(x,y),y$2) = 0;u:=(x,y) -> arctan(y/x);
> LaplaceEqn;
> simplify(LaplaceEqn);
```

$$0 = 0$$

■ A unique solution to a PDE is obtained by using an appropriate number of initial conditions (conditions dependent on time given typically at $t = 0$), and/or boundary conditions (conditions at specific points of the solution domain known as boundaries of the domain).

Example 4 -- Verify that $u(x,y) = a \ln(x^2+y^2)+b$ satisfies Laplace's equation in two dimensions, and determine a and b so that u satisfies the boundary condition $u = 0$ on the circle $x^2+y^2 = 1$ and $u = 5$ on the circle $x^2+y^2 = 9$.

To verify that the function $u(x,y)$ satisfies the Laplace equation use:

```
> restart:LaplaceEqn:=diff(u(x,y),x$2)+diff(u(x,y),y$2) = 0;
```

> u:=(x,y)->a*ln(x^2+y^2)+b;simplify(LaplaceEqn);

$$0 = 0$$

To evaluate the boundary conditions use:

> u:='u':Definition:=u=a*ln(x^2+y^2)+b;

$$\text{Definition} := u = a \ln(x^2 + y^2) + b$$

> eq1:=subs({x^2+y^2=1,u=0},Definition);eq2:=subs({x^2+y^2=9,u=5},Definition);

$$\text{eq1} := 0 = a \ln(1) + b$$

$$\text{eq2} := 5 = a \ln(9) + b$$

> solve({eq1,eq2},{a,b});evalf(%);

$$\left\{ a = 5 \frac{1}{\ln(9)}, b = 0 \right\} \quad \left\{ a = 2.275598067, b = 0 \right\}$$

■ In general there should be as many boundary or initial conditions as the highest order of the corresponding partial derivative. For example, the one dimensional heat equation (equation [2]) applied to a insulated bar of length L , will require an initial condition, say

$$u(x, t=0) = f(x), \quad 0 < x < L,$$

as well as two boundary conditions, e.g., $u(x=0, t) = u_0$ and $u(x=L, t) = u_L$, or, $u_x(x=0, t) = u_{x0}$ and $u_x(x=L, t) = u_{xL}$, or some combination of these, for $t > 0$.

Classification of linear, second-order PDEs

Linear, second-order PDEs, as the examples shown above as equations [1] through [4], are commonly encountered in science and engineering applications. For that reason special attention is paid in this section to this type of equations. First, we learn how to classify linear, second-order PDEs as follows:

An equation of the form:

$$Au_{xx} + 2Bu_x + Cu_{yy} = F(x, y, u, u_x, u_y),$$

is said to be:

■ parabolic, if $AC - B^2 = 0$, e.g., heat flow and diffusion-type problems.

■ hyperbolic, if $AC - B^2 < 0$, e.g., vibrating systems and wave motion problems.

■ elliptic, if $AC - B^2 > 0$, e.g., steady-state, potential-type problems.

Analytical solutions of PDEs

There are a variety of methods for obtaining symbolic, or closed-form, solutions to differential equations. In this section we will present two of those techniques:

- Separation of variables
- Integral transforms (Laplace transforms, Fourier transforms)

The method of separation of variables can be used to obtain analytical solutions for some simple PDEs. The method consists in writing the general solution as the product of functions of a single variable, then replacing the resulting function into the PDE, and separating the PDE into ODEs of a single variable each. The ODEs are solved separately and their solutions combined into the solution of the PDE.

In many cases, the ODEs resulting from the separation of variables produce solutions that depend on a parameter known as an *eigenvalue* (if the eigenvalue appears in a sine or cosine function that depends on time, it is referred to as an *eigenfrequency*). The solutions involving eigenvalues are known as *eigenfunctions*.

The use of integral transforms reduces a PDE in n independent variables into one that has only $(n-1)$ independent variables, thus a PDE with two variables can be easily changed into an ODE.

Integral Transforms

In Chapter 13 we introduced Laplace transforms for the solution of ordinary differential equations. In this section we introduce other integral transforms that we will use in the solution of partial differential equations. In general, an *integral transform* is a transformation that relates a function $f(t)$ to a new function $F(s)$ by an integration of the form

$$F(s) = \int_a^b \kappa(s, t) \cdot f(t) \cdot dt.$$

The function $\kappa(s, t)$ is known as the *kernel of the transformation*.

The use of an integral transform allows us to resolve a function into a given *spectrum of components*. To understand the concept of a spectrum, consider the Fourier series

$$f(t) = a_0 + \sum_{n=1}^{\infty} (a_n \cdot \cos \omega_n x + b_n \cdot \sin \omega_n x),$$

representing a periodic function with a period T . This Fourier series can be re-written as

$$f(x) = a_0 + \sum_{n=1}^{\infty} A_n \cdot \cos\left(\frac{n\pi x}{T} + \phi_n\right),$$

where

$$A_n = \sqrt{a_n^2 + b_n^2}, \quad \tan^{-1}\left(\frac{b_n}{a_n}\right)$$

for $n = 1, 2, \dots$

The amplitudes A_n will be referred to as the spectrum of the function and will be a measure of the magnitude of the component of $f(x)$ with frequency

$$f_n = n/T.$$

The basic or fundamental frequency in the Fourier series is $f_0 = 1/T$, thus, all other frequencies are multiples of this basic frequency, i.e., $f_n = n \cdot f_0$. Also, we can define an angular frequency,

$$\omega_n = 2\pi n/T = 2\pi \cdot f_n = 2\pi \cdot n \cdot f_0 = n \cdot \omega_0,$$

where ω_0 is the basic or fundamental angular frequency of the Fourier series.

Using the angular frequency notation, the Fourier series expansion is written as

$$f(x) = a_0 + \sum_{n=1}^{\infty} A_n \cdot \cos(\omega_n x + \phi_n) = a_0 + \sum_{n=1}^{\infty} (a_n \cdot \cos \omega_n x + b_n \cdot \sin \omega_n x)$$

A plot of the values A_n vs. ω_n is the typical representation of a discrete spectrum for a function. The discrete spectrum will show that the function has components at angular frequencies ω_n which are integer multiples of the fundamental angular frequency ω_0 .

Suppose that we are faced with the need to expand a non-periodic function into sine and cosine components. A non-periodic function can be thought of as having an infinitely large period. Thus, for a very large value of T , the fundamental angular frequency, $\omega_0 = 2\pi/T$, becomes a very small quantity, say $\Delta\omega$. Also, the angular frequencies corresponding to $\omega_n = n \cdot \omega_0 = n \cdot \Delta\omega$, ($n = 1, 2, \dots, \infty$), now take values closer and closer to each other, suggesting the need for a continuous spectrum of values.

The non-periodic function can be written, therefore, as

$$f(x) = \int_0^{\infty} [C(\omega) \cdot \cos(\omega \cdot x) + S(\omega) \cdot \sin(\omega \cdot x)] d\omega,$$

where

$$C(\omega) = \frac{1}{2\pi} \cdot \int_{-\infty}^{\infty} f(x) \cdot \cos(\omega \cdot x) \cdot dx,$$

and

$$S(\omega) = \frac{1}{2\pi} \cdot \int_{-\infty}^{\infty} f(x) \cdot \sin(\omega \cdot x) \cdot dx.$$

The continuous spectrum is given by

$$A(\omega) = \sqrt{[C(\omega)]^2 + [S(\omega)]^2}$$

The functions $C(\omega)$, $S(\omega)$, and $A(\omega)$ are continuous functions of a variable ω , which becomes the transform variable for the Fourier transforms defined below.

Example 1 - Determine the coefficients $C(\omega)$, $S(\omega)$, and the continuous spectrum $A(\omega)$, for the function $f(x) = \exp(-x)$, for $x > 0$, and $f(x) = 0$, $x < 0$.

Using Maple, the value of $C(\omega)$ is found with:

```
> '(1/sqrt(2*Pi))*int(exp(-x)*cos(omega*x),x=0..infinity)';C:=%;
```

$$\frac{\int_0^{\infty} e^{-x} \cos(\omega x) dx}{\sqrt{2\pi}} \quad C := \frac{1}{2} \frac{\sqrt{2}}{\sqrt{\pi} (1 + \omega^2)}$$

Also, the calculation of $S(\omega)$ is found with:

```
> '(1/sqrt(2*Pi))*int(exp(-x)*sin(omega*x),x=0..infinity)';S:=%;
```

$$\frac{\int_0^{\infty} e^{-x} \sin(\omega x) dx}{\sqrt{2\pi}} \quad S := \frac{1}{2} \frac{\sqrt{2} \omega}{\sqrt{\pi} (1 + \omega^2)}$$


The continuous spectrum, $A(\omega)$ is calculated as:

```
> A:=sqrt(C^2+S^2);simplify(A);
```

$$\frac{1}{2} \frac{\sqrt{2}}{\sqrt{1 + \omega^2} \sqrt{\pi}}$$

Define this expression as a function and plot the spectrum:

```
> A:=omega -> 1/2*sqrt(2)/(sqrt(1+omega^2)*sqrt(Pi));plot(A(omega),omega = 0..10);
```



Fourier transforms

In Chapter 13 we introduced the Laplace transforms indicating that the transform takes a function, $f(t)$, in t -space, and transform it into a function, $F(s)$, in s -space. Thus, $F(s) = \mathcal{L}\{f(t)\}$ represents the Laplace transform of $f(t)$, and $f(t) = \mathcal{L}^{-1}\{F(s)\}$ represents the inverse Laplace transform of $F(s)$. In this section we introduce three types of transform and their inverses as follows:

■ Fourier sine transform:

$$F_s\{f(t)\} = F(\omega) = \frac{2}{\pi} \cdot \int_0^{\infty} f(t) \cdot \sin(\omega \cdot t) \cdot dt$$

■ Inverse sine transform:

$$F_s^{-1}\{F(\omega)\} = f(t) = \int_0^{\infty} F(\omega) \cdot \sin(\omega \cdot t) \cdot dt$$

■ Fourier cosine transform:

$$F_c\{f(t)\} = F(\omega) = \frac{2}{\pi} \cdot \int_0^{\infty} f(t) \cdot \cos(\omega \cdot t) \cdot dt$$

■ Inverse cosine transform:

$$F_c^{-1}\{F(\omega)\} = f(t) = \int_0^{\infty} F(\omega) \cdot \cos(\omega \cdot t) \cdot dt$$

■ Fourier transform (proper):

$$F\{f(t)\} = F(\omega) = \frac{1}{\sqrt{2\pi}} \cdot \int_{-\infty}^{\infty} f(t) \cdot e^{-i\omega t} \cdot dt$$

■ Inverse Fourier transform

$$F^{-1}\{F(\omega)\} = f(t) = \frac{1}{\sqrt{2\pi}} \cdot \int_{-\infty}^{\infty} F(\omega) \cdot e^{i\omega t} \cdot dt$$

Example 1 - Determine the Fourier transform of the function $f(t) = \exp(-t)$, for $t > 0$, and $f(t) = 0$, for $t < 0$.

The continuous spectrum, $F(\omega)$, is calculated with the integral:

$$\frac{1}{\sqrt{2\pi}} \int_0^{\infty} e^{-(1+i\omega)t} dt = \lim_{\varepsilon \rightarrow \infty} \frac{1}{\sqrt{2\pi}} \int_0^{\varepsilon} e^{-(1+i\omega)t} dt = \lim_{\varepsilon \rightarrow \infty} \frac{1}{\sqrt{2\pi}} \left[\frac{1 - \exp(-(1+i\omega)t)}{1+i\omega} \right] = \frac{1}{\sqrt{2\pi}} \cdot \frac{1}{1+i\omega}.$$

This result can be obtained in Maple by using:

```
> '(1/sqrt(2*Pi))*int(exp(-(1+I*omega)*t),t=0..infinity)';F:=%;
```

$$\frac{\int_0^{\infty} e^{-(1+I\omega)t} dt}{\sqrt{2\pi}} \quad F := \frac{1}{2} \frac{\sqrt{2}}{\sqrt{\pi} (1+I\omega)}$$

This result can be rationalized by multiplying numerator and denominator by the conjugate of the denominator, namely, $1-i\omega$. The result is now:

$$F(\omega) = \frac{1}{\sqrt{2\pi}} \cdot \frac{1}{1+i\omega} = \frac{1}{\sqrt{2\pi}} \cdot \left(\frac{1}{1+i\omega} \right) \cdot \left(\frac{1-i\omega}{1-i\omega} \right) = \frac{1}{\sqrt{2\pi}} \left(\frac{1}{1+\omega^2} - i \cdot \frac{\omega}{1+\omega^2} \right)$$


In Maple, this is accomplished by using:

```
> evalc(F);
```

$$\frac{1}{2} \frac{\sqrt{2}}{\sqrt{\pi} (1+\omega^2)} - \frac{1}{2} \frac{I\sqrt{2}\omega}{\sqrt{\pi} (1+\omega^2)}$$

The result just obtained is the complex function $F(\omega)$. The real and imaginary parts of the function can be plotted as shown below:

```
> ReF:=Re(F);ImF:=Im(F);plot({ReF,ImF},omega = 0..5);
```



Notes:

1. The absolute value of the Fourier transform, $|F(\omega)|$, is the frequency spectrum of the original function $f(t)$. For the example shown above, $|F(\omega)| = 1/[2\pi(1+\omega^2)]^{1/2}$. The plot of $|F(\omega)|$ vs. ω was shown earlier.
2. Some functions, such as constant values, $\sin x$, $\exp(x)$, x^2 , etc., do not have Fourier transform. Functions that go to zero sufficiently fast as x goes to infinity do have Fourier transforms.

Properties of the Fourier transform

Let Φ represent a Fourier transform, i.e., a sine, cosine, or full Fourier transform. The transforms possess the following properties:

1. *Linearity*: If a and b are constants, and f and g functions, then $\Phi\{a \cdot f + b \cdot g\} = a \Phi\{f\} + b \Phi\{g\}$.
2. *Transformation of partial derivatives*. Let $u = u(x, t)$. If the Fourier transform transforms the variable x , then
 - $\Phi\{\partial u / \partial x\} = i\omega \Phi\{u\}$
 - $\Phi\{\partial^2 u / \partial x^2\} = -\omega^2 \Phi\{u\}$
 - $\Phi\{\partial u / \partial t\} = \partial \Phi\{u\} / \partial t$
 - $\Phi\{\partial^2 u / \partial t^2\} = \partial^2 \Phi\{u\} / \partial t^2$
3. *Convolution*: For Fourier transform applications, the operation of convolution is defined as

$$(f * g)(x) = \frac{1}{\sqrt{2\pi}} \cdot \int f(x - \xi) \cdot g(\xi) \cdot d\xi.$$

The following property holds for convolution:

$$\Phi\{f * g\} = \Phi\{f\} \cdot \Phi\{g\}.$$

Fourier transforms and convolution using Maple

Maple provides the functions *fouriersin*, *fouriercos*, *fourier*, and *invfourier*, within package *inttrans* (*integral transforms*), to calculate, respectively, the Fourier sine, Fourier cosine, Fourier transform, and inverse Fourier transforms. These transforms are defined slightly differently than the definitions provided before. The Maple definitions for the transforms are as follows:

■ Fourier sine transform:

$$F_s\{f(t)\} = F(\omega) = \sqrt{\frac{2}{\pi}} \cdot \int_0^{\infty} f(t) \cdot \sin(\omega \cdot t) \cdot dt$$

■ Fourier cosine transform:

$$F_c\{f(t)\} = F(\omega) = \sqrt{\frac{2}{\pi}} \cdot \int_0^{\infty} f(t) \cdot \cos(\omega \cdot t) \cdot dt$$

■ Fourier transform:

$$F\{f(t)\} = F(\omega) = \int_{-\infty}^{\infty} f(t) \cdot e^{-i\omega t} \cdot dt$$

■ Inverse Fourier transform

$$F^{-1}\{F(\omega)\} = f(t) = \frac{1}{2\pi} \cdot \int_{-\infty}^{\infty} F(\omega) \cdot e^{i\omega t} \cdot dt$$

Although not included in the list of functions in package *inttrans*, the inverse Fourier sine and cosine transforms, following the definitions of the Fourier sine and cosine transforms, would be calculated as:

■ Inverse sine transform:

$$F_s^{-1}\{F(\omega)\} = f(t) = \sqrt{\frac{2}{\pi}} \cdot \int_0^{\infty} F(\omega) \cdot \sin(\omega \cdot t) \cdot dt$$

■ Inverse cosine transform:

$$F_c^{-1}\{F(\omega)\} = f(t) = \sqrt{\frac{2}{\pi}} \cdot \int_0^{\infty} F(\omega) \cdot \cos(\omega \cdot t) \cdot dt$$

Some examples of Fourier transforms using the functions of package *inttrans* are shown next:

Example 1 - Determine the Fourier sine, Fourier cosine, and Fourier transforms of the function $f(t) = 1/t$:

> restart:with(inttrans):f:=t->1/(1+t);

$$f := t \rightarrow \frac{1}{1+t}$$

> fouriersin(f(t),t,omega);

$$\frac{\sqrt{2} (\sin(\omega) \text{Ci}(\omega) - \cos(\omega) \text{Ssi}(\omega))}{\sqrt{\pi}}$$

> fouriercos(f(t),t,omega);

$$\frac{\sqrt{2} (-\sin(\omega) \text{Ssi}(\omega) - \cos(\omega) \text{Ci}(\omega))}{\sqrt{\pi}}$$

> fourier(f(t),t,omega);

$$I e^{(I \omega)} \pi (\text{Heaviside}(-\omega) - \text{Heaviside}(\omega))$$

Example 2. Use Fourier transforms to obtain the convolution of the functions $f(t) = t$ and $g(t) = \exp(-t^2)$:

> restart:with(inttrans):f:=t->t:g:=t->exp(-t^2);

$$f := t \rightarrow t$$

$$g := t \rightarrow e^{(-t^2)}$$

> F:=omega->fourier(f(t),t,omega):G:=omega->fourier(g(t),t,omega):F(omega);G(omega);

$$2 I \pi \text{Dirac}(1, \omega)$$

$$\sqrt{\pi} e^{\left(-\frac{1}{4} \omega^2\right)}$$

> H:=omega->F(omega)*G(omega):H(omega);

$$2 I \pi \begin{pmatrix} 3 \\ 2 \end{pmatrix} \text{Dirac}(1, \omega) e^{\left(-\frac{1}{4} \omega^2\right)}$$

> invfourier(H(omega),omega,t);

$$\sqrt{\pi t}$$

Analytical solutions to parabolic equations: one-dimensional solution of the heat equation

The flow of heat in a thin, laterally insulated homogeneous rod is modeled by

$$\partial u / \partial t = k \cdot (\partial^2 u / \partial x^2),$$


where u = temperature, k = a parameter resulting from combining thermal conductivity and density, together with an initial condition

$$u(x, 0) = f(x)$$

and constant-value boundary conditions

$$u(0, t) = u_0, \text{ and } u(L, t) = u_L.$$

The physical phenomenon described by this PDE and its initial and boundary conditions is illustrated in the figure below with $u_0 = u_L = 0$.



In Maple, we can enter the heat equation as:

```
> restart;
> heat_eqn:= diff(u(x,t),t) = k*diff(u(x,t),x$2);
```

$$heat_eqn := \frac{\partial}{\partial t} u(x, t) = k \left(\frac{\partial^2}{\partial x^2} u(x, t) \right)$$

We will try to find a solution by the method of separation of variables. This method assumes that the solution, $u(x,t)$, can be expressed as the product of two functions, $X(x)$ and $T(t)$, as follows:

```
> u:= (x,t) -> X(x)*T(t);
```

$$u := (x, t) \rightarrow X(x) T(t)$$

With this substitution, the initial condition, $u(x,t=0) = f(x) = X(x)T(t)$, can be treated as the set of conditions: $X(x) = f(x)$, when $t = 0$ [i.e., $T(t) = 1$]. Also, the boundary conditions, $u(0,t) = X(0)T(t) = u[0]$; and $u(L,t) = X(L)T(t) = u[L]$; can be treated as $X(0) = u[0]$, and $X(L) = u[L]$; as long as $T(t) \neq 0$;

The derivatives of $u(x,t)$ are calculated as follows:

```
> 'diff(u(x,t),x$2)'; diff(u(x,t),x$2);
```

$$\frac{\partial^2}{\partial x^2} u(x, t) \quad \left(\frac{\partial^2}{\partial x^2} X(x) \right) T(t)$$

```
> 'diff(u(x,t),t)'; diff(u(x,t),t);
```

$$\frac{\partial}{\partial t} u(x, t) \quad X(x) \left(\frac{\partial}{\partial t} T(t) \right)$$

Replacing these derivatives in the heat equation we get:

```
> heat_eqn;
```

$$X(x) \left(\frac{\partial}{\partial t} T(t) \right) = k \left(\frac{\partial^2}{\partial x^2} X(x) \right) T(t)$$

Dividing by $u(x,t) = X(x)T(t)$:

```
> heat_eqn:= heat_eqn/u(x,t);
```

$$heat_eqn := \frac{\frac{\partial}{\partial t} T(t)}{T(t)} = \frac{k \left(\frac{\partial^2}{\partial x^2} X(x) \right)}{X(x)}$$

This result is only possible if both sides of the equation are equal to a constant, say $-\alpha$, since the left-hand side is only a function of t and the right hand side is only function of x . The left-hand side of the heat equation produces an ODE with independent variable t :

```
> lhs_eqn:= diff(T(t),t)/T(t)=-alpha;
```

$$lhs_eqn := \frac{\frac{\partial}{\partial t} T(t)}{T(t)} = -\alpha$$

Whose solution is:

```
> dsolve({lhs_eqn,T(0)=1},T(t));
```

$$T(t) = e^{(-\alpha t)}$$

On the other hand, the right-hand side of the heat equation produces an ODE with independent variable is x :

```
> rhs_eqn:=k*diff(X(x),`$`(x,2))/X(x)=-alpha;
```

$$rhs_eqn := \frac{k \left(\frac{\partial^2}{\partial x^2} X(x) \right)}{X(x)} = -\alpha$$

Let's try to solve this equation directly using the boundary conditions $u[0] = u[L] = 0$:

```
> dsolve({rhs_eqn,X(0) = 0, X(L) = 0}, X(x));
```

$$X(x) = 0$$

We find a trivial solution by direct integration. Let's try a different approach:

```
> rhs_sol:= dsolve(rhs_eqn, X(x));
```

$$rhs_sol := X(x) = _C1 \sin\left(\sqrt{\frac{\alpha}{k}} x\right) + _C2 \cos\left(\sqrt{\frac{\alpha}{k}} x\right)$$

Next, we replace the boundary conditions:

```
> eqCA:= evalf(subs({X(x) = 0, x = 0}, rhs_sol));
```

$$\text{eqCA} := 0 = 1 \cdot _C2$$

> $_C2 := \text{solve}(\text{eqCA}, _C2);$

$$_C2 := 0$$

$_C2$ turns out to be equal to zero, simplifying the solution to:

> $\text{rhs_sol};$

$$X(x) = _C1 \sin\left(\sqrt{\frac{\alpha}{k}} x\right)$$

> $\text{eqCB} := \text{evalf}(\text{subs}(\{X(x) = 0, x = L\}, \text{rhs_sol}));$

$$\text{eqCB} := 0 = _C1 \sin\left(L \sqrt{\frac{\alpha}{k}}\right)$$

The latter result indicates an eigenfunction problem, we need to find all possible values of α ; for which this equation is satisfied. Since we want $_C1 \neq 0$, then we set

$$\sin\left(L \sqrt{\frac{\alpha}{k}}\right) = 0$$

> $\text{solve}(\sin(L \cdot \sqrt{\alpha/k})=0, \{\alpha\});$

$$\{\alpha = 0\}$$

The function solve gives only the first solution to the equation. We know, however, that the sine function has multiple solutions located at,

$$L \sqrt{\frac{\alpha}{k}} = \dots - 3\pi, -2\pi, -\pi, 0, \pi, 2\pi, 3\pi, \dots$$

i.e.,

> $\text{eqn_for_alpha} := L \cdot \sqrt{\alpha/k} = n \cdot \pi; \# \text{ for } n = \dots -3, -2, -1, 0, 1, 2, 3, \dots$

$$\text{eqn_for_alpha} := L \sqrt{\frac{\alpha}{k}} = n \pi$$

> $\alpha := \text{solve}(\text{eqn_for_alpha}, \alpha);$

$$\alpha := \frac{n^2 \pi^2 k}{L^2}$$

Therefore, with these values of α the solution for *rhs_eqn* now becomes:
 > rhs_sol;

$$X(x) = _C1 \sin\left(\sqrt{\frac{n^2 \pi^2}{L^2}} x\right)$$

The value of $_C1$ remains somewhat arbitrary, requiring a different approach to find it. To simplify notation we will replace $_C1$ with b_n :

> subs($_C1=b[n]$, rhs_sol);

$$X(x) = b_n \sin\left(\sqrt{\frac{n^2 \pi^2}{L^2}} x\right)$$

With the value of found earlier, the solution to *lhs_eqn* is now:

> subs(alpha= $n^2 \pi^2 k / (L^2)$, T(t) = exp(-alpha*t));

$$T(t) = e^{\left(-\frac{n^2 \pi^2 k t}{L^2}\right)}$$

There will be a different expression for $u(x,t) = X(t)T(n)$ for each value of $n = \dots -3, -2, -1, 0, 1, 2, 3, \dots$. Therefore, we will call the solution corresponding to a particular value of n $u_n(x,t)$ and write:

> u_sub_n := (x,t) -> b[n]*sin((n*Pi/L)*x)*exp(-n^2*Pi^2*k*t/(L^2));

$$u_{sub_n} := (x, t) \rightarrow b_n \sin\left(\frac{n \pi x}{L}\right) e^{\left(-\frac{n^2 \pi^2 k t}{L^2}\right)}$$

The form of the n -th solution, u_n , suggests an expansion similar to a Fourier series expansion for the overall solution, $u(x,t)$, restricting the values of n to positive integers, i.e.,

> u := (x,t) -> sum(u_sub_n(x,t), n = 1..infinity);#resulting expression

$$u := (x, t) \rightarrow \sum_{n=1}^{\infty} u_{sub_n}(x, t)$$

> u(x,t);

$$\sum_{n=1}^{\infty} \sin\left(\frac{n \pi x}{L}\right) b_n = f(x)$$

with the values of b_n obtained from the boundary condition, $u(x,0) = f(x)$, i.e.,
 > $u(x,0) = f(x)$;

$$\sum_{n=1}^{\infty} b_n \sin\left(\frac{n \pi x}{L}\right) e^{\left(-\frac{n^2 \pi^2 k t}{L^2}\right)}$$

The latter result is a Fourier sine series with the coefficients $b[n]$; given by:

> $b := n \rightarrow (2/L) * \text{int}(f(x) * \sin(n * \text{Pi} * x / L), x = 0..L)$;

$$b := n \rightarrow 2 \frac{\int_0^L f(x) \sin\left(\frac{n \pi x}{L}\right) dx}{L}$$

Example 1 - Determine the solution for the one-dimensional heat equation subjected to $u(0,t) = u(L,t) = 0$, if the initial conditions are given by $u(x,0) = f(x) = 4 \cdot (x/L)^3 \cdot (1-x/L)^2$.

Without loss of generality we can take $L = 1$, and define the function:


> restart;

> $f := x \rightarrow 4 * x^3 * (1-x)^2$;

$$f := x \rightarrow 4 x^3 (1 - x)^2$$

A plot of the function is shown below:

> plot(f(x), x = 0..1);



The Fourier series coefficients are calculated as:

> b:= n -> 2*int(f(x)*sin(n*Pi*x), x = 0..1);

$$b := n \rightarrow 2 \int_0^1 f(x) \sin(n \pi x) dx$$

Generating 10 components:

> co1:= seq(b(n), n = 1..10);

$$\text{co1} := -16 \frac{-12 + \pi^2}{\pi^5}, 2 \frac{-15 + \pi^2}{\pi^5}, -\frac{16}{81} \frac{-4 + 3 \pi^2}{\pi^5}, \frac{1}{16} \frac{-15 + 4 \pi^2}{\pi^5}, -\frac{16}{3125} \frac{-12 + 25 \pi^2}{\pi^5}, \frac{2}{81} \frac{-5 + 3 \pi^2}{\pi^5},$$

$$-\frac{16}{16807} \frac{-12 + 49 \pi^2}{\pi^5}, \frac{1}{512} \frac{-15 + 16 \pi^2}{\pi^5}, -\frac{16}{19683} \frac{-4 + 27 \pi^2}{\pi^5}, \frac{2}{625} \frac{-3 + 5 \pi^2}{\pi^5}$$

> evalf(co1);#Floating-point values of the coefficients

.1113860683, -.03352984038, -.01653008369, .004999355193, -.003927425008, .001985574939,
-.001467114439, .0009121251879, -.0006972273534, .0004846540198


Assuming k = 1, the solution u(x,t) will be expressed by the function F(x,t,n) defined as:

> F:= (x,t,n) -> sum(co1[j]*sin(j*Pi*x)*exp(-j^2*Pi^2*t), j = 1..n);

$$F := (x, t, n) \rightarrow \sum_{j=1}^n \text{co1}_j \sin(j \pi x) e^{(-j^2 \pi^2 t)}$$


Plot of the temperature distribution at t = 0.1:

```
> plot(F(x,0.1,10), x = 0..1);
```



A time-lapse surface of the solution $u(x,t)$ can be constructed using:

```
> plot3:= plot3d(F(x,t,10), x = 0..1, t = 0..0.1);  
> plots[display](plot3, style = patch, axes = frame);
```



A different view of the solution surface is achieved by using:

```
>plots[display](plot3, style = patch, axes = frame, orientation=[-117, 63]);
```


Slices of the surface solution, i.e., curves of $u(x,t)$ vs. x for different values of t , are shown next:

```
> ts:= [seq(0.01*(j-1), j = 1..11)]; #Selected values of time
```

```
    ts := [0, .01, .02, .03, .04, .05, .06, .07, .08, .09, .10]
```

```
> many_plots:= {seq(plot(F(x,ts[r],10),x=0..1),r = 1..11)};
```

```
> plots[display](many_plots);
```



To see an animation of the function $F(x,t)$, which will show F vs. x for different values of t , use:

```
> plots[animate]( F(x,t,10), x=0..1, t=0..0.5, frames=50, color = blue);
```

Try this command in Maple to see the animated graph.

Analytical solutions to hyperbolic equations: One-dimensional solution of the wave equation

The wave equation, shown below, can be used to model the displacement of an elastic string or the longitudinal vibration of a beam:

```
> restart:WaveEqn:=diff(u(x,t),t$2) = c^2*diff(u(x,t),x$2);
```

$$\text{WaveEqn} := \frac{\partial^2}{\partial t^2} u(x, t) = c^2 \left(\frac{\partial^2}{\partial x^2} u(x, t) \right)$$

For the case of an elastic string, $c^2 = \frac{T}{\mu}$, where T is the constant tension in the string and μ is the mass per unit length of the string. For the case of longitudinal vibration of a beam, $c^2 = \frac{gE}{\rho}$, where g is the acceleration of gravity, E is the modulus of elasticity, and ρ is the density of the beam.

Suppose we solve the wave equation for a vibrating string of length L using separation of variables with the boundary conditions $u(0, t) = u(L, 0) = 0$. Also, the initial shape of the string is given by $u(x, 0) = f(x)$, and the initial speed of the string is given by $\frac{\partial}{\partial t} u(x, t) = g(x)$ at $t = 0$. We postulate a solution of the form $u(x, t) = X(x)T(t)$, i.e.,

```
> u:=(x,t) -> X(x)*T(t);
```

$$u := (x, t) \rightarrow X(x) T(t)$$

and replace this result in the original PDE:

```
> WaveEqn;
```

$$X(x) \left(\frac{\partial^2}{\partial t^2} T(t) \right) = c^2 \left(\frac{\partial^2}{\partial x^2} X(x) \right) T(t)$$

Dividing both sides of the equation by $u(x, t)$, we get:

```
> WaveEqn1:=simplify(WaveEqn/u(x,t));
```

$$\text{WaveEqn1} := \frac{\frac{\partial^2}{\partial t^2} T(t)}{T(t)} = \frac{c^2 \left(\frac{\partial^2}{\partial x^2} X(x) \right)}{X(x)}$$

This equation is only possible if the two sides of the equations are equal to a constant, say, $-\alpha^2$. With this we can write two ODEs, one for each side of the equation WaveEqn1 , i.e.,

```
> ODE1:=diff(T(t),t$2)+alpha^2*T(t)=0; ODE2:=diff(X(x),x$2)+(alpha/c)^2*X(x)=0;
```

$$\text{ODE1} := \left(\frac{\partial^2}{\partial t^2} T(t) \right) + \alpha^2 T(t) = 0$$

$$\text{ODE2} := \left(\frac{\partial^2}{\partial x^2} X(x) \right) + \frac{\alpha^2 X(x)}{c^2} = 0$$

The solutions to these equations are:

```
> Sol1:=dsolve(ODE1,T(t));Sol2:=dsolve(ODE2,X(x));
```

$$\begin{aligned} \text{Sol1} &:= T(t) = _C1 \cos(\alpha t) + _C2 \sin(\alpha t) \\ \text{Sol2} &:= X(x) = _C1 \sin\left(\frac{\alpha x}{c}\right) + _C2 \cos\left(\frac{\alpha x}{c}\right) \end{aligned}$$

Note: the constants $_C1$ and $_C2$ in the two solutions are not the same.

The boundary conditions $u(0,t) = u(L,t) = 0$ translate into $X(0) = X(L) = 0$. With these boundary conditions, we can form the following algebraic equations:

$$\begin{aligned} > \text{Eq1Sol2} := \text{simplify}(\text{subs}(\{x=0, X(x)=0\}, \text{Sol2})); \\ & \text{Eq1Sol2} := 0 = _C2 \\ > _C2 := 0; \text{Eq2Sol2} := \text{simplify}(\text{subs}(\{x=L, X(x)=0\}, \text{Sol2})); \\ & _C2 := 0 \\ & \text{Eq2Sol2} := 0 = _C1 \sin\left(\frac{\alpha L}{c}\right) \end{aligned}$$

We have an eigenvalue equation given by $\sin\left(\frac{\alpha L}{c}\right) = 0$. The solution to this equation is $\frac{\alpha L}{c} = 0, \pi, 2\pi, \dots$, or $\alpha = 0, \frac{\pi c}{L}, \frac{2\pi c}{L}, \dots$. Using only the positive values, we can write $\alpha = \frac{n\pi c}{L}$, $n = 0, 1, 2, \dots$. Next, using the value of α and $_C2 = 1$, we obtain an expression for $X(x)$ as follows:

$$\begin{aligned} > \text{alpha} := n * \text{Pi} * c / L; \text{Sol2} := \text{subs}(\{_C1=1\}, \text{Sol2}); \text{assign}(\%); \\ & \alpha := \frac{n\pi c}{L} \\ & \text{Sol2} := X(x) = \sin\left(\frac{n\pi x}{L}\right) \end{aligned}$$

function $T(t)$ gets written as (we need to redefine $_C1$ and $_C2$ first):

$$\begin{aligned} > _C1 := _C1'; _C2 := _C2'; \text{Sol1}; \\ & T(t) = _C1 \cos\left(\frac{n\pi c t}{L}\right) + _C2 \sin\left(\frac{n\pi c t}{L}\right) \\ > T := t \rightarrow _C1 * \cos(n * \text{Pi} * c * t / L) + _C2 * \sin(n * \text{Pi} * c * t / L); \\ & T := t \rightarrow _C1 \cos\left(\frac{n\pi c t}{L}\right) + _C2 \sin\left(\frac{n\pi c t}{L}\right) \end{aligned}$$

The function $u(x,t)$ is now:

$$> u(x,t);$$

$$\sin\left(\frac{n\pi x}{L}\right) \left(_C1 \cos\left(\frac{n\pi c t}{L}\right) + _C2 \sin\left(\frac{n\pi c t}{L}\right) \right)$$

Application of the initial conditions provide the following equations:

$$\begin{aligned} > \text{EqSol}_1 := f(x) = u(x,0); \\ & \text{EqSol}_1 := f(x) = _C1 \sin\left(\frac{n\pi x}{L}\right) \\ > \text{EqSol}_2 := g(x) = \text{simplify}(\text{subs}(t=0, \text{diff}(u(x,t), t))); \\ & \text{EqSol}_2 := g(x) = \frac{\sin\left(\frac{n\pi x}{L}\right) _C2 n\pi c}{L} \end{aligned}$$

Since the last two equations need to apply for $n = 0, 1, 2, \dots$, we recognize in them the equations that define Fourier cosine series if we use $_C1 = a_n$, and $_C2 = b_n$, i.e.,

```

> _C1:=a[n];_C2:=b[n];EqSol_1;EqSol_2;

      f(x) = a_n sin\left(\frac{n \pi x}{L}\right)

      g(x) = \frac{\sin\left(\frac{n \pi x}{L}\right) b_n n \pi c}{L}

> Eqa:=int(a[n]*sin(n*Pi*x/L)^2,x=0..L)=int(f(x)*sin(n*Pi*x/L),x=0..L);

      Eqa := -\frac{1}{2} \frac{a_n L (\cos(n \pi) \sin(n \pi) - n \pi)}{n \pi} = \int_0^L f(x) \sin\left(\frac{n \pi x}{L}\right) dx
      -\frac{1}{2} b_n c \cos(n \pi) \sin(n \pi) + \frac{1}{2} b_n c n \pi

> Eqb:=int(g(x)*sin(n*Pi*x/L),x=0..L)=int(b[n]*n*Pi*c/L*sin(n*Pi*x/L)^2,x=0..L);

      Eqb := \int_0^L g(x) \sin\left(\frac{n \pi x}{L}\right) dx = -\frac{1}{2} b_n c \cos(n \pi) \sin(n \pi) + \frac{1}{2} b_n c n \pi

> assume(n,integer);Eqa;Eqb;

      \frac{1}{2} a_n L = \int_0^L f(x) \sin\left(\frac{n \pi x}{L}\right) dx

      \int_0^L g(x) \sin\left(\frac{n \pi x}{L}\right) dx = \frac{1}{2} b_n c n \pi

```

The equations defining coefficients a_n and b_n are given by:

$$a_n = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{n \pi x}{L}\right) dx, \text{ and } b_n = \frac{2}{n \pi c} \int_0^L g(x) \sin\left(\frac{n \pi x}{L}\right) dx, \text{ for } n = 1, 2, 3, \dots$$

The final solution is, therefore,

```

> u:='u':u:=(x,t)->sum(sin(n*Pi*x/L)*(a[n]*cos(n*Pi*c*t/L)+b[n]*sin(n*Pi*c*t/L)),n=1..infinity);u(x,t);

      u := (x, t) \to \sum_{n=1}^{\infty} \sin\left(\frac{n \pi x}{L}\right) \left( a_n \cos\left(\frac{n \pi c t}{L}\right) + b_n \sin\left(\frac{n \pi c t}{L}\right) \right)

```

$$\sum_{n=1}^{\infty} \sin\left(\frac{n\pi x}{L}\right) \left(a_n \cos\left(\frac{n\pi c t}{L}\right) + b_n \sin\left(\frac{n\pi c t}{L}\right) \right)$$

Example 1 - Consider the case of a vibrating string with the initial displacement given by $f(x) = \frac{x}{L} \left(1 - \frac{x}{L}\right)$, and the initial velocity given by $g(x) = \left(\frac{x}{L}\right)^2 \left(1 - \frac{x}{L}\right)$. The boundary conditions are $u(0, t) = 0$, $u(L, t) = 0$. Determine the solution $u(x, t)$ for this problem using components of the resulting Fourier series for $n = 1, 2, \dots, 20$, if $c = 1$ and $L = 1$.

```
> restart:f:=x->(x/L)*(1-x/L);g:=x->(x/L)^2*(1-x/L);#Define functions f(x) and g(x)
```

$$f := x \rightarrow \frac{x \left(1 - \frac{x}{L}\right)}{L}$$

$$g := x \rightarrow \frac{x \left(1 - \frac{x}{L}\right)^2}{L}$$

The following commands define the coefficients of the Fourier series, a_n and b_n :

```
> a:=n->(2/L)*int(f(x)*sin(n*Pi*x/L),x=0..L);
```

$$a := n \rightarrow 2 \frac{\int_0^L f(x) \sin\left(\frac{n\pi x}{L}\right) dx}{L}$$

```
> b:=n->(2/(n*Pi*c))*int(g(x)*sin(n*Pi*x/L),x=0..L);
```

$$b := n \rightarrow 2 \frac{\int_0^L g(x) \sin\left(\frac{n\pi x}{L}\right) dx}{n\pi c}$$

A list of the first 20 coefficients follows:

```
> aa:=seq(a(n),n=1..20);bb:=seq(b(n),n=1..20);
```

$$aa := 8 \frac{1}{\pi^3}, 0, \frac{8}{27} \frac{1}{\pi^3}, 0, \frac{8}{125} \frac{1}{\pi^3}, 0, \frac{8}{343} \frac{1}{\pi^3}, 0, \frac{8}{729} \frac{1}{\pi^3}, 0, \frac{8}{1331} \frac{1}{\pi^3}, 0, \frac{8}{2197} \frac{1}{\pi^3}, 0, \frac{8}{3375} \frac{1}{\pi^3}, 0, \frac{8}{4913} \frac{1}{\pi^3}, 0, \frac{8}{6859} \frac{1}{\pi^3}, 0$$

$$bb := 8 \frac{1}{\pi^4}, 0, \frac{8}{81} \frac{1}{\pi^4}, 0, \frac{8}{625} \frac{1}{\pi^4}, 0, \frac{8}{2401} \frac{1}{\pi^4}, 0, \frac{8}{6561} \frac{1}{\pi^4}, 0, \frac{8}{14641} \frac{1}{\pi^4}, 0, \frac{8}{28561} \frac{1}{\pi^4}, 0, \frac{8}{50625} \frac{1}{\pi^4}, 0, \frac{8}{83521} \frac{1}{\pi^4}, 0, \frac{8}{130321} \frac{1}{\pi^4}, 0$$

The approximate solution using 20 components is given by:

```
> u:=(x,t)->sum(sin(n*Pi*x/L)*(aa[n]*cos(n*Pi*c*t/L)+bb[n]*sin(n*Pi*c*t/L)),n=1..20);
```

$$u := (x, t) \rightarrow \sum_{n=1}^{20} \sin\left(\frac{n\pi x}{L}\right) \left(aa_n \cos\left(\frac{n\pi c t}{L}\right) + bb_n \sin\left(\frac{n\pi c t}{L}\right) \right)$$


Defining parameters for this problem:

□

```
> c:=1;L:=1;plot(f(x),x=0..L);plot(g(x),x=0..L);
```


$c := 1$

$L := 1$




A three-dimensional plot of the signal $u(x,t)$ follows:

```
> plot3d(u(x,t),x=0..1,t=0..4,axes = boxed);
```



An animation of the signal is given by:

```
> plots[animate]( u(x,t), x=0..1, t=0..4,frames=50, color = blue, scaling = constrained);
```



Analytical solutions to elliptic equations: Two-dimensional solution to Laplace's equation in a rectangular domain.

Laplace's equation in two-dimensions is given by

$$\left(\frac{\partial^2}{\partial x^2} u(x, y)\right) + \left(\frac{\partial^2}{\partial y^2} u(x, y)\right) = 0.$$


In problems related to heat transfer, the two-dimensional Laplace equation describes the steady state distribution of temperature $u(x, y)$ in the x - y plane. In fluid mechanics, $u(x, y)$ could describe the velocity potential or the streamfunction for a two-dimensional potential flow. The problem requires two boundary conditions in the two independent variables x and y .

Solution to Case [1].

Laplace equation is solved in a rectangular domain so that $0 < x < L$, $0 < y < H$, a suitable set of boundary conditions may be

$$u(x, 0) = 0, u(x, H) = g(x), u(0, y) = 0, u(L, y) = 0.$$

as illustrated in the figure below:



Separation of variables suggests that we use a solution of the form,

$$u(x, y) = X(x) Y(y).$$

Solving the equation through separation of variables proceeds in the following fashion:

```
> restart; LEq:=diff(u(x,y),x$2)=-diff(u(x,y),y$2);u:=(x,y) -> X(x)*Y(y);LEq;
```

$$LEq := \frac{\partial^2}{\partial x^2} u(x, y) = - \left(\frac{\partial^2}{\partial y^2} u(x, y) \right)$$

$$u := (x, y) \rightarrow X(x) Y(y)$$

$$\left(\frac{\partial^2}{\partial x^2} X(x) \right) Y(y) = -X(x) \left(\frac{\partial^2}{\partial y^2} Y(y) \right)$$

Dividing by $u(x, y) = X(x)Y(y)$ results in:

```
> LEq:=simplify(LEq/u(x,y));
```

$$LEq := \frac{\frac{\partial^2}{\partial x^2} X(x)}{X(x)} = - \frac{\frac{\partial^2}{\partial y^2} Y(y)}{Y(y)}$$

For the two sides of the resulting equation to be equal, they both must be equal to a constant, λ^2 , resulting in the following two ordinary differential equations:

> ODE1:=diff(X(x),x\$2)+lambda^2*X(x)=0;ODE2:=diff(Y(y),y\$2)-lambda^2*Y(y)=0;

$$ODE1 := \left(\frac{\partial^2}{\partial x^2} X(x) \right) + \lambda^2 X(x) = 0$$

$$ODE2 := \left(\frac{\partial^2}{\partial y^2} Y(y) \right) - \lambda^2 Y(y) = 0$$

The solution to the first ODE is:

> Sol1:=dsolve(ODE1,X(x));

$$Sol1 := X(x) = _C1 \sin(\lambda x) + _C2 \cos(\lambda x)$$

We next use the boundary conditions: $X(0) = 0$, and $X(L) = 0$, to determine the constants of integration:

> Eq1:=simplify(subs({x=0,X(x)=0},Sol1));

$$Eq1 := 0 = _C2$$

> _C1:=0;#Assign value of zero to _C1

$$_C1 := 0$$

> Eq2:=simplify(subs({x=L,X(x)=0},Sol1));

$$Eq2 := 0 = _C2 \cos(\lambda L)$$

This second equation produces an eigenvalue equation with the eigenvalues given by $\lambda = \frac{n \pi}{L}$.

> lambda:=n*Pi/L;

$$\lambda := \frac{n \pi}{L}$$

The solution for $X(x)$, with $_C2 = 1$ (since the value $_C2$ is arbitrary), is, therefore:

> _C2:=1;Sol1;

$$_C2 := 1$$

$$X(x) = \cos\left(\frac{n \pi x}{L}\right)$$

The solution to the second ODE is:

> Sol2:=dsolve(ODE2,Y(y));

$$Sol2 := Y(y) = _C3 \cosh\left(\frac{n \pi y}{L}\right) + _C4 \sinh\left(\frac{n \pi y}{L}\right)$$

Utilizing the boundary condition: $Y(0) = 0$, we find for $Y(y)$:

> Eq3:=simplify(subs({y=0,Y(y)=0},Sol2));

$$Eq3 := 0 = _C3$$

> _C3:=0;Sol2;

$$_C3 := 0$$

$$Y(y) = C_4 \sinh\left(\frac{n \pi y}{L}\right)$$

The product $X(x)Y(y)$, which now depends on the value of the eigenvalue $\lambda_n = \frac{n \pi}{L}$, is referred to as

$v_n(x, y) = X(x) Y(y)$, with the constant replaced by a_n :

```
> v[n]:= (x,y) -> a[n]*sin(n*Pi*x/L)*sinh(n*Pi*y/L);
```

$$v_n := (x, y) \rightarrow a_n \sin\left(\frac{n \pi x}{L}\right) \sinh\left(\frac{n \pi y}{L}\right)$$

The solution will be the sum of all possible functions $v_n(x, y)$, i.e.,

```
> u:=(x,y) -> sum(v[n](x,y),n=1..infinity);
```

$$u := (x, y) \rightarrow \sum_{n=1}^{\infty} v_n(x, y)$$

The expression for the function $u(x, y)$ is:

```
> u(x,y);
```

$$\sum_{n=1}^{\infty} a_n \sin\left(\frac{n \pi x}{L}\right) \sinh\left(\frac{n \pi y}{L}\right)$$

If we now evaluate the boundary condition, $u(x, H) = g(x)$, we find the following equation:

```
> Eq4:=u(x,H) = g(x);
```

$$Eq4 := \sum_{n=1}^{\infty} a_n \sin\left(\frac{n \pi x}{L}\right) \sinh\left(\frac{n \pi H}{L}\right) = g(x)$$

This result is a Fourier series expansion to $g(x)$, such that the constants a_n are calculated by

```
> a:='a';a:=n->2*int(g(x)*sin(n*Pi*x/L),x=0..L)/(L*sinh(n*Pi*H/L));
```

$$a := n \rightarrow 2 \frac{\int_0^L g(x) \sin\left(\frac{n \pi x}{L}\right) dx}{L \sinh\left(\frac{n \pi H}{L}\right)}$$


Example 1: Suppose that the dimensions of the solution domain are $L = 2$ and $H = 1$, and the boundary condition at $y = H$ is given by $g(x) = 100 x (L - x)$. A plot of the boundary condition follows:

```
> H:=1;L:=2;g:=x->100*x*(L-x)^3;plot(subs(L=1,g(x)),x=0..1);
```

$$H := 1$$

$$L := 2$$

$$g := x \rightarrow 100 x (L - x)^3$$



The first twenty coefficients a_n are calculated and stored in the sequence called *coef*:

```
> coef:=seq(evalf(a(n)),n=1..20);
coef := 50.97163411, 6.702349078, .3749726169, .03613686210, .003721486167, .0004626992502,
        .00005956977515, .8435410605 10-5, .1219249529 10-5, .1866378986 10-6, .2895439597 10-7, .4667450766 10-8,
        .7594815776 10-9, .1270173419 10-9, .2139034491 10-10, .3677145602 10-11, .6355007076 10-12, .1116032657 10-12,
        .1968199894 10-13, .3515831712 10-14
```

We put together the function $u(x, y)$ as a series of 20 terms:


```
> u:='u';u:=(x,y) -> sum(coef[j]*sin(j*Pi*x/L)*sinh(j*Pi*y/L),j=1..20);
```

$$u := u$$

$$u := (x, y) \rightarrow \sum_{j=1}^{20} coef_j \sin\left(\frac{j \pi x}{L}\right) \sinh\left(\frac{j \pi y}{L}\right)$$


The following is a three-dimensional plot of the surface $u(x, y)$

```
> plot3d(u(x,y),x=0..L,y=0..H,axes = boxed);
```



A contour plot of $u(x,y)$ is shown below.


```
> plots[contourplot](u(x,y),x=0..L,y=0..H,contours=20,axes=boxed,
filled=true,coloring=[red,blue],scaling=constrained);
```



The solution obtained above was facilitated by the use of zero boundary conditions in three of the boundaries.

The zero boundary conditions at $x = 0$ and $x = L$ produced the eigenvalues $\lambda_n = \frac{n\pi}{L}$, while the zero boundary condition at $y = 0$ produced the series solution $u(x, y) = \sum_{n=1}^{\infty} a_n \sin\left(\frac{n\pi x}{L}\right) \sinh\left(\frac{n\pi y}{L}\right)$. Finally, the non-zero boundary condition at $y = H$, produced the coefficients a_n for the corresponding Fourier series.

Solutions to the case of a rectangular domain where only one of the boundaries is non-zero can be found for cases [2] through [4] in the figure below using a procedure similar to that outlined above for case [1].



Because Laplace's equation is a linear equation, i.e., $\nabla^2(\phi_1 + \phi_2) = \nabla^2\phi_1 + \nabla^2\phi_2$, solutions with different boundary conditions, in the same domain, can be superimposed. Therefore, linear combinations of the solutions for the four cases illustrated in the figure above can be added to solve problems involving non-zero boundary conditions in more than one boundary. Suppose that $u_1(x, y)$, $u_2(x, y)$, $u_3(x, y)$, and $u_4(x, y)$, represent the solution for cases [1],[2],[3], and [4], respectively, in the figure above. The most general problem in the rectangular domain will use the following non-zero boundary conditions:

$$u(x, 0) = \alpha_1 g(x), u(x, H) = \alpha_2 f(x), u(0, y) = \alpha_3 p(y), \text{ and } u(L, y) = \alpha_4 q(y).$$

The solution is obtained as

$$u(x, y) = \alpha_1 u_1(x, y) + \alpha_2 u_2(x, y) + \alpha_3 u_3(x, y) + \alpha_4 u_4(x, y).$$

Solution to case [2].

```
> restart;LEq:=diff(u(x,y),x$2)=-diff(u(x,y),y$2);u:=(x,y) -> X(x)*Y(y);LEq;
```

$$LEq := \frac{\partial^2}{\partial x^2} u(x, y) = - \left(\frac{\partial^2}{\partial y^2} u(x, y) \right)$$

$$u := (x, y) \rightarrow X(x) Y(y)$$

$$\left(\frac{\partial^2}{\partial x^2} X(x) \right) Y(y) = -X(x) \left(\frac{\partial^2}{\partial y^2} Y(y) \right)$$

Dividing by $u(x,y) = X(x)Y(y)$ results in:

```
> LEq:=simplify(LEq/u(x,y));
```

$$LEq := \frac{\frac{\partial^2}{\partial x^2} X(x)}{X(x)} = - \frac{\frac{\partial^2}{\partial y^2} Y(y)}{Y(y)}$$

For the two sides of the resulting equation to be equal, they both must be equal to a constant, λ^2 , resulting in the following two ordinary differential equations:

```
> ODE1:=diff(X(x),x$2)+lambda^2*X(x)=0;ODE2:=diff(Y(y),y$2)-lambda^2*Y(y)=0;
```

$$ODE1 := \left(\frac{\partial^2}{\partial x^2} X(x) \right) + \lambda^2 X(x) = 0$$

$$ODE2 := \left(\frac{\partial^2}{\partial y^2} Y(y) \right) - \lambda^2 Y(y) = 0$$

The solution to the first ODE is:

```
> Sol1:=dsolve(ODE1,X(x));
```

$$Sol1 := X(x) = _C1 \sin(\lambda x) + _C2 \cos(\lambda x)$$

We next use the boundary conditions: $X(0) = 0$, and $X(L) = 0$, to determine the constants of integration:

```
> Eq1:=simplify(subs({x=0,X(x)=0},Sol1));
```

$$Eq1 := 0 = _C2$$

```
> _C1:=0;# Assign value of zero to _C1
```

$$_C1 := 0$$

```
> Eq2:=simplify(subs({x=L,X(x)=0},Sol1));
```

$$Eq2 := 0 = _C2 \cos(\lambda L)$$

This second equation produces an eigenvalue equation with the eigenvalues given by $\lambda = \frac{n \pi}{L}$.

```
> lambda:=n*Pi/L;
```

$$\lambda := \frac{n \pi}{L}$$

The solution for $X(x)$, with $_C2 = 1$ (since the value $_C2$ is arbitrary), is, therefore:

```
> _C2:=1;Sol1;
```

$$_C2 := 1$$

$$X(x) = \cos\left(\frac{n\pi x}{L}\right)$$

The solution to the second ODE is:

> Sol2:=dsolve(ODE2,Y(y));

$$\text{Sol2} := Y(y) = _C3 \cosh\left(\frac{n\pi y}{L}\right) + _C4 \sinh\left(\frac{n\pi y}{L}\right)$$

Utilizing the boundary condition: $Y(H) = 0$, we find for $Y(y)$:

> Eq3:=simplify(subs({y=H,Y(y)=0},Sol2));

$$\text{Eq3} := 0 = _C3 \cosh\left(\frac{n\pi H}{L}\right) + _C4 \sinh\left(\frac{n\pi H}{L}\right)$$

> _C4:=solve(Eq3,_C4);factor(Sol2);

$$_C4 := -\frac{_C3 \cosh\left(\frac{n\pi H}{L}\right)}{\sinh\left(\frac{n\pi H}{L}\right)}$$

$$Y(y) = -\frac{_C3 \left(-\cosh\left(\frac{n\pi y}{L}\right) \sinh\left(\frac{n\pi H}{L}\right) + \cosh\left(\frac{n\pi H}{L}\right) \sinh\left(\frac{n\pi y}{L}\right) \right)}{\sinh\left(\frac{n\pi H}{L}\right)}$$

The product $X(x)Y(y)$, which now depends on the value of the eigenvalue $\lambda_n = \frac{n\pi}{L}$, is referred to as

$v_n(x, y) = X(x) Y(y)$, with the constant replaced by a_n :

> v[n]:= (x,y) ->

$a[n] \sin(n\pi x/L) (\cosh(n\pi y/L) \sinh(n\pi H/L) - \cosh(n\pi H/L) \sinh(n\pi y/L)) / \sinh(n\pi H/L);$

$$v_n := (x, y) \rightarrow \frac{a_n \sin\left(\frac{n\pi x}{L}\right) \left(\cosh\left(\frac{n\pi y}{L}\right) \sinh\left(\frac{n\pi H}{L}\right) - \cosh\left(\frac{n\pi H}{L}\right) \sinh\left(\frac{n\pi y}{L}\right) \right)}{\sinh\left(\frac{n\pi H}{L}\right)}$$

The solution will be the sum of all possible functions $v_n(x, y)$, i.e.,

> u:= (x,y) -> sum(v[n](x,y),n=1..infinity);

$$u := (x, y) \rightarrow \sum_{n=1}^{\infty} v_n(x, y)$$

The expression for the function $u(x, y)$ is:

> u(x,y);

$$\sum_{n=1}^{\infty} \frac{a_n \sin\left(\frac{n\pi x}{L}\right) \left(\cosh\left(\frac{n\pi y}{L}\right) \sinh\left(\frac{n\pi H}{L}\right) - \cosh\left(\frac{n\pi H}{L}\right) \sinh\left(\frac{n\pi y}{L}\right) \right)}{\sinh\left(\frac{n\pi H}{L}\right)}$$

If we now evaluate the boundary condition, $u(x, 0) = g(x)$, we find the following equation:

> Eq4:=u(x,0) = f(x);

$$Eq4 := \sum_{n=1}^{\infty} a_n \sin\left(\frac{n \pi x}{L}\right) = f(x)$$


This result is a Fourier series expansion to $f(x)$, such that the constants a_n are calculated by

```
> a:='a';a:=n->(2/L)*int(f(x)*sin(n*Pi*x/L),x=0..L);
```

$$a := n \rightarrow 2 \frac{\int_0^L f(x) \sin\left(\frac{n \pi x}{L}\right) dx}{L}$$

Example 1: Suppose that the dimensions of the solution domain are $L = 2$ and $H = 1$, and the boundary condition at $y = H$ is given by $f(x) = 100 x (L - x)$. A plot of the boundary condition follows:

```
> H:=1;L:=2;f:=x->100*x*(L-x)^3;plot(subs(L=1,f(x)),x=0..1);
```



The first twenty coefficients a_n are calculated and stored in the sequence called *coef*:

```
> coef:=seq(evalf(a(n)),n=1..20);
```

```
coef := 117.3009656, 77.40368261, 20.86887512, 9.675460326, 4.793218569, 2.866803060, 1.775469452,
1.209432541, .8409229256, .6192294609, .4621196780, .3583503824, .2805004413, .2256667131, .1828144201,
.1511790676, .1256854639, .1061778911, .09007713623, .07740368261
```

We put together the function $u(x, y)$ as a series of 20 terms:


```
> u:='u';u:=(x,y) ->
```

```
sum(coef[j]*sin(j*Pi*x/L)*(cosh(j*Pi*y/L)*sinh(j*Pi*H/L)-cosh(j*Pi*H/L)*sinh(j*Pi*y/L))/sinh(j*Pi*H/L),
j=1..20);
```

$$u := (x, y) \rightarrow \sum_{j=1}^{20} \frac{\text{coef}_j \sin\left(\frac{j \pi x}{L}\right) \left(\cosh\left(\frac{j \pi y}{L}\right) \sinh\left(\frac{j \pi H}{L}\right) - \cosh\left(\frac{j \pi H}{L}\right) \sinh\left(\frac{j \pi y}{L}\right) \right)}{\sinh\left(\frac{j \pi H}{L}\right)}$$


The following is a three-dimensional plot of the surface $u(x, y)$

```
> plot3d(u(x,y),x=0..L,y=0..H,axes = boxed);
```




A contour plot of $u(x, y)$ is shown below.

```
> plots[contourplot](u(x,y),x=0..L,y=0..H,contours=20,axes=boxed,
filled=true,coloring=[red,blue],scaling=constrained);
```



Superposition of solutions for cases [1] and [2]

The superposition of the solutions for cases [1] and [2] will be the solution to the problem whose boundary conditions are illustrated in the figure below:



We use for the solution $f(x) = x(L-x)^3$, and $g(x) = x^3(L-x)$.

Next, we reproduce the individual solutions as $u_1(x, y)$ and $u_2(x, y)$, respectively, as indicated below. First, the solution to case [1] is:

```

> restart:L:=2;H:=1;f:=x->x*(L-x)^3;g:=x->x^3*(L-x);

                                L := 2
                                H := 1
                                f := x → x (L - x)3
                                g := x → x3 (L - x)

> a1:=n->2*int(g(x)*sin(n*Pi*x/L),x = 0 .. L)/(L*sinh(n*Pi*H/L)):coef1:=seq(evalf(a1(j)),j=1..20);
coef1 := .5097163411, -.06702349078, .003749726169, -.0003613686210, .00003721486167, -.4626992502 10-5,
.5956977515 10-6, -.8435410605 10-7, .1219249529 10-7, -.1866378986 10-8, .2895439597 10-9, -.4667450766 10-10,
.7594815776 10-11, -.1270173419 10-11, .2139034491 10-12, -.3677145602 10-13, .6355007076 10-14,
-.1116032657 10-14, .1968199894 10-15, -.3515831712 10-16

> u1:=(x,y) -> sum(coef1[j]*sin(j*Pi*x/L)*sinh(j*Pi*y/L),j=1..20);u1(x,y):

                                u1 := (x, y) → ∑j=120 coef1j sin(  $\frac{j \pi x}{L}$  ) sinh(  $\frac{j \pi y}{L}$  )

```

Next, the solution to case [2] follows:

```

> a2:=n->(2/L)*int(f(x)*sin(n*Pi*x/L),x = 0 .. L):coef2:=seq(evalf(a2(j)),j=1..20);
coef2 := 1.173009656, .7740368261, .2086887512, .09675460326, .04793218569, .02866803060, .01775469452,
.01209432541, .008409229256, .006192294609, .004621196780, .003583503824, .002805004413, .002256667131,
.001828144201, .001511790676, .001256854639, .001061778911, .0009007713623, .0007740368261

> u2:=(x,y) ->
sum(coef2[j]*sin(j*Pi*x/L)*(cosh(j*Pi*y/L)*sinh(j*Pi*H/L)-cosh(j*Pi*H/L)*sinh(j*Pi*y/L))/sinh(j*Pi*H/L),
j = 1..20);u2(x,y):

                                u2 := (x, y) → ∑j=120  $\frac{coef2_j \sin\left(\frac{j \pi x}{L}\right) \left( \cosh\left(\frac{j \pi y}{L}\right) \sinh\left(\frac{j \pi H}{L}\right) - \cosh\left(\frac{j \pi H}{L}\right) \sinh\left(\frac{j \pi y}{L}\right) \right)}{\sinh\left(\frac{j \pi H}{L}\right)}$ 

```

Combining the two solutions:

```

> u:=u1(x,y)+u2(x,y):


```

The three-dimensional plot of the surface $z = u(x, y)$ is shown next:

```


> plot3d(u,x=0..L,y=0..H,axes = boxed);

```



The corresponding contour plot is:

```
> plots[contourplot](u(x,y),x=0..L,y=0..H,contours=20,axes=boxed,
filled=true,coloring=[red,blue],scaling=constrained);
```




Suppose now, that the boundary conditions are such that $f_1(x) = 3 f(x) = 3 x (L - x)^3$ and $g_1(x) = 5 g(x) = 5 x^3 (L - x)$. Since we already have the solutions corresponding to $f(x)$ and $g(x)$, we can obtain the solution to the problem with the new set of boundary conditions by constructing the following solution

$u(x, y) = 3 u_1(x, y) + 5 u_2(x, y)$:


```
[ > u:=3*u1(x,y)+5*u2(x,y):
```

The surface and contour plots corresponding to this solution are shown below:

```
> plot3d(u,x=0..L,y=0..H,axes = boxed);
```



```
> plots[contourplot](u(x,y),x=0..L,y=0..H,contours=20,axes=boxed,
filled=true,coloring=[red,blue],scaling=constrained);
```



Laplacian in polar-cylindrical coordinates

Consider the function $u = u(r, \theta, z)$ in polar coordinates. The laplacian of this function in polar coordinates can be obtained from Maple using:

```
> restart:with(linalg):laplacian(u(r,theta,z),[r,theta,z],coords=cylindrical);
Warning, new definition for norm
Warning, new definition for trace
```

$$\frac{\left(\frac{\partial}{\partial r} u(r, \theta, z)\right) + r \left(\frac{\partial^2}{\partial r^2} u(r, \theta, z)\right) + \frac{\frac{\partial^2}{\partial \theta^2} u(r, \theta, z)}{r} + r \left(\frac{\partial^2}{\partial z^2} u(r, \theta, z)\right)}{r}$$

You can check that the first two terms are equivalent to $\frac{\partial}{\partial r} r \left(\frac{\partial}{\partial r} u(r, \theta, z)\right)$, i.e.,

```
> diff(r*diff(u(r,theta,z),r),r);
```

$$\left(\frac{\partial}{\partial r} u(r, \theta, z)\right) + r \left(\frac{\partial^2}{\partial r^2} u(r, \theta, z)\right)$$

Thus, the Laplacian of $u(x,y,z)$ can be written as:


$$\nabla^2 u(r, \theta, z) = \frac{1}{r} \left(\frac{\partial}{\partial r} r \left(\frac{\partial}{\partial r} u(r, \theta, z) \right) \right) + \frac{1}{r^2} \frac{\partial^2}{\partial \theta^2} u(r, \theta, z) + \frac{\partial^2}{\partial z^2} u(r, \theta, z)$$

In the next section we will discuss the solution of the Laplace equation in two dimensions using polar-cylindrical coordinates, the corresponding Laplacian, taking $u = u(r, \theta)$ and $\frac{\partial}{\partial z} u = 0$, is

$$\nabla^2 u(r, \theta, z) = \frac{1}{r} \left(\frac{\partial}{\partial r} r \left(\frac{\partial}{\partial r} u(r, \theta) \right) \right) + \frac{1}{r^2} \frac{\partial^2}{\partial \theta^2} u(r, \theta)$$

Solution to Laplace's equation in an annulus

Consider the domain of solution to Laplace equation to be an annulus defined, in polar coordinates, as $r_1 < r < r_2$, and $0 < \theta < 2\pi$, subjected to the boundary conditions: $u(r_1, \theta) = u_1 = \text{constant}$, and $u(r_2, \theta) = u_2 = \text{constant}$, as illustrated in the figure below:



In polar coordinates, Laplace's equation is written as: $\frac{1}{r} \left(\frac{\partial}{\partial r} r \left(\frac{\partial}{\partial r} u(r, \theta) \right) \right) + \frac{1}{r^2} \frac{\partial^2}{\partial \theta^2} u(r, \theta) = 0$. If the solution is assumed to depend only on the radial coordinate, $u = u(r)$, and $\frac{\partial}{\partial \theta} u = 0$. Thus, Laplace's equation reduces to $\frac{1}{r} \frac{\partial}{\partial r} r \left(\frac{\partial}{\partial r} u(r) \right) = 0$. This is indeed an ordinary differential equation and we could write it as $\frac{1}{r} \frac{d}{dr} \left(r \frac{du}{dr} \right) = 0$, or $\frac{d}{dr} \left(r \frac{du}{dr} \right) = 0$. Integration of this equation produces:

```
> restart:LEQP:=diff(r*diff(u(r),r),r) = 0;SOL:=dsolve(LEQP,u(r));
      LEQP := (∂/∂r u(r)) + r (∂²/∂r² u(r)) = 0
      SOL := u(r) = _C1 + _C2 ln(r)
```


Using the boundary conditions we obtain:

```
> SOL:=dsolve({LEQP,u(r_1)=u_1,u(r_2)=u_2},u(r));
      SOL := u(r) = (-ln(r_2) u_1 + u_2 ln(r_1)) / (-ln(r_2) + ln(r_1)) - (u_2 - u_1) ln(r) / (-ln(r_2) + ln(r_1))
```

For example, if

```
> u_1:=1;r_1:=1;u_2:=10;r_2:=5;SOL;plot(rhs(SOL),r=1..5);
      u_1 := 1
```

$$\begin{aligned}
 r_1 &:= 1 \\
 u_2 &:= 10 \\
 r_2 &:= 5 \\
 u(r) &= 1 + 9 \frac{\ln(r)}{\ln(5)}
 \end{aligned}$$



Solution to the Laplace equation in a disk

Consider the case in which $u = u(r, \theta)$ as solution to the Laplace equation on a disk centered at the origin and subjected to the boundary condition, $u(R, \theta) = f(\theta)$, where R is the radius of the disk. Using separation of variables we can write $u(r, \theta) = R(r) T(\theta)$, requiring that $T(\theta)$ be periodic in θ with period 2π . Thus, we can write $T(\theta)$ as a Fourier series, i.e.,

$$T(\theta) = \frac{a_0}{2} + \left(\sum_{k=1}^{\infty} (a_k \cos(k\theta) + b_k \sin(k\theta)) \right)$$

The procedure for the solution using separation of variables proceeds as follows: The Laplace equation in cylindrical coordinates is written as:

```
> restart;LEQP:=r*diffr(diffr(u(r,theta),r),r)=-difu(r,theta),theta$2);
```

$$LEQP := r \left(\left(\frac{\partial}{\partial r} u(r, \theta) \right) + r \left(\frac{\partial^2}{\partial r^2} u(r, \theta) \right) \right) = - \left(\frac{\partial^2}{\partial \theta^2} u(r, \theta) \right)$$

Separating the equation into ordinary differential equations we use:

```
> u:=(r,theta) -> R(r)*T(theta);LEQPS:=simplify(LEQP/u(r,theta));
```

$$u := (r, \theta) \rightarrow R(r) T(\theta)$$

$$LEQPS := \frac{r \left(\left(\frac{\partial}{\partial r} R(r) \right) + r \left(\frac{\partial^2}{\partial r^2} R(r) \right) \right)}{R(r)} = - \frac{\frac{\partial^2}{\partial \theta^2} T(\theta)}{T(\theta)}$$

The differential equations are:

```
> ODE1:=r*(differ(R(r),r)+r*differ(R(r),`$` (r,2))) -lambda^2*R(r)=0;Sol1:=dsolve(ODE1,R(r));
```

$$ODE1 := r \left(\left(\frac{\partial}{\partial r} R(r) \right) + r \left(\frac{\partial^2}{\partial r^2} R(r) \right) \right) - \lambda^2 R(r) = 0$$

$$Sol1 := R(r) = _C1 \sinh(\lambda \ln(r)) + _C2 \cosh(\lambda \ln(r))$$

Since the expression $\lambda \ln(r)$ is equivalent to $\ln(r^\lambda)$, we can replace the following results:

$$\begin{aligned} &> _C1*\sinh(\ln(r^\lambda))+_C2*\cosh(\ln(r^\lambda)); \\ &\qquad\qquad\qquad -C1\left(\frac{1}{2}r^\lambda - \frac{1}{2}\frac{1}{r^\lambda}\right) + -C2\left(\frac{1}{2}r^\lambda + \frac{1}{2}\frac{1}{r^\lambda}\right) \end{aligned}$$

And re-write Sol1:

$$\begin{aligned} &> \text{Sol1}:=R(r) = _C1*r^\lambda + _C2*r^{(-\lambda)}; \\ &\qquad\qquad\qquad \text{Sol1} := R(r) = _C1 r^\lambda + _C2 r^{(-\lambda)} \end{aligned}$$

Typically, the values of λ are positive values, therefore, the term $r^{(-\lambda)}$ grows unbounded as $r \rightarrow 0$. This requires that we eliminate the term $r^{(-\lambda)}$ from the solution by making $_C2 = 0$, thus:

$$\begin{aligned} &> _C2:=0;\text{Sol1}; \\ &\qquad\qquad\qquad _C2 := 0 \\ &\qquad\qquad\qquad R(r) = _C1 r^\lambda \end{aligned}$$

The second ODE and its solution are:

$$\begin{aligned} &> \text{ODE2}:= -\text{diff}(T(\theta),\theta) - \lambda^2 * T(\theta) = 0; \text{Sol2}:=\text{dsolve}(\text{ODE2},T(\theta)); \\ &\qquad\qquad\qquad \text{ODE2} := -\left(\frac{\partial^2}{\partial \theta^2} T(\theta)\right) - \lambda^2 T(\theta) = 0 \\ &\qquad\qquad\qquad \text{Sol2} := T(\theta) = _C1 \sin(\lambda \theta) + _C3 \cos(\lambda \theta) \end{aligned}$$

The latter result indicates that the initial assumption of a periodic function for $T(\theta)$ is acceptable if we take $\lambda = k$. We now replace Sol2 with:

$$\begin{aligned} &> \text{Sol2}:=T(\theta) = a[0]/2 + \text{sum}(a[k]*\cos(k*\theta) + b[k]*\sin(k*\theta), k = 1 .. \text{infinity}); \\ &\qquad\qquad\qquad \text{Sol2} := T(\theta) = \frac{1}{2} a_0 + \left(\sum_{k=1}^{\infty} (a_k \cos(k \theta) + b_k \sin(k \theta)) \right) \end{aligned}$$

The combined solution, $u(r, \theta) = R(r) T(\theta)$, is now:

$$\begin{aligned} &> u:='u'; u:=(r,\theta) \rightarrow 1/2*a[0] + \text{sum}(r^k*(a[k]*\cos(k*\theta) + b[k]*\sin(k*\theta)), k = 1 .. \text{infinity}); \\ &\qquad\qquad\qquad u := u \\ &\qquad\qquad\qquad u := (r, \theta) \rightarrow \frac{1}{2} a_0 + \left(\sum_{k=1}^{\infty} r^k (a_k \cos(k \theta) + b_k \sin(k \theta)) \right) \end{aligned}$$

The boundary conditions, $u(R, \theta) = f(\theta)$, produce:

$$\begin{aligned} &> \text{EqBC}:=u(R,\theta) = f(\theta); \\ &\qquad\qquad\qquad \text{EqBC} := \frac{1}{2} a_0 + \left(\sum_{k=1}^{\infty} R^k (a_k \cos(k \theta) + b_k \sin(k \theta)) \right) = f(\theta) \end{aligned}$$

With the coefficients of the resulting Fourier series given by:

$$a_0 = \frac{1}{\pi} \int_0^{2\pi} f(\theta) d\theta, \quad a_k = \frac{1}{\pi R^k} \int_0^{2\pi} f(\theta) \cos(k \theta) d\theta, \quad \text{and} \quad b_k = \frac{1}{\pi R^k} \int_0^{2\pi} f(\theta) \sin(k \theta) d\theta.$$

For example, if $R = 10$, and $f(\theta) = \sin(\theta) - 2.5 \sin(2 \theta) + 1.5 \cos(3 \theta)$:

$$\begin{aligned} &> R:=10; f:=\theta \rightarrow \sin(\theta) + 2.5*\sin(2*\theta) + 1.5*\cos(3*\theta); \\ &\qquad\qquad\qquad R := 10 \end{aligned}$$

$$f := \theta \rightarrow \sin(\theta) + 2.5 \sin(2\theta) + 1.5 \cos(3\theta)$$

The coefficients for the solution are defined as:

```
> a0:=(1/Pi)*int(f(theta),theta=0..2*Pi);a:=k->(1/(Pi*R^k))*int(f(theta)*cos(k*theta),theta=0..2*Pi);b:=k->(1/(Pi*R^k))*int(f(theta)*sin(k*theta),theta=0..2*Pi);
```

$$a_0 := 0$$

$$a := k \rightarrow \frac{\int_0^{2\pi} f(\theta) \cos(k\theta) d\theta}{\pi R^k}$$

$$b := k \rightarrow \frac{\int_0^{2\pi} f(\theta) \sin(k\theta) d\theta}{\pi R^k}$$

The coefficients for the cosine series and sine series are *cc* and *cs*, respectively.

```
> cc:=seq(a(k),k=1..20);cs:=seq(b(k),k=1..20);
```

$$cc := 0, 0, .004712388981 \frac{1}{\pi}, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0$$

$$cs := .3141592654 \frac{1}{\pi}, .07853981635 \frac{1}{\pi}, 0$$

The solution is:

```
> u:='u';u:=(r,theta) -> 1/2*a0+sum(r^k*(cc[k]*cos(k*theta)+cs[k]*sin(k*theta)),k = 1 .. 20);
```

$$u := (r, \theta) \rightarrow \frac{1}{2} a_0 + \left(\sum_{k=1}^{20} r^k (cc_k \cos(k\theta) + cs_k \sin(k\theta)) \right)$$

To produce a three-dimensional plot first we create lists of values of *r* (*rList*) and *theta* (*thetaList*) dividing the corresponding ranges, $0 < r < R$, $0 < \theta < 2\pi$, into $n = 25$ parts:

```
> n:=5;m:=20;rList:=seq(j*R/n,j=0..n);thetaList:=seq(2*Pi*j/m,j=0..m);
```


$$n := 5$$

$$m := 20$$

Next, we create a list of lists for points $[r, \theta, u(r, \theta)]$ and plot it using function *surfdata* in package *plots*. The resulting surface shows the solution $u(r, \theta)$:

```
> pointList:=seq([seq([rList[i]*cos(thetaList[j]),rList[i]*sin(thetaList[j]),u(rList[i],thetaList[j])],j=1..m)],i=1..n);
```

```
> with(plots):surfdata(pointList,axes=boxed,labels=["x","y","u"]);
```



The *PDEtools* package

The *PDEtools* package contains a number of functions useful in the solution of partial differential equations. For example, the function *pdsolve* will attempt to obtain a solution for a PDE as illustrated in the example below:

```
[ > restart:with(PDEtools): # I invoke package PDEtools
  > PDE1:=diff(u(x,y),x$2)=c^2*diff(u(x,y),y$2);#The wave equation
```

$$PDE1 := \frac{\partial^2}{\partial x^2} u(x, y) = c^2 \left(\frac{\partial^2}{\partial y^2} u(x, y) \right)$$

```
> Sol1:=pdsolve(PDE1,u(x,y));
```

$$Sol1 := u(x, y) = _F2(\sqrt{c^2} x + y) + _F1\left(\frac{1}{2}x - \frac{1}{2}\frac{y}{\sqrt{c^2}}\right)$$

The result provided for the wave equation represents a general solution in the sense that any two functions *_F1* and *_F2* of the given arguments will satisfy the equation. The package *PDEtools* provides function *pdetest* that can be used to check if an expression satisfy a given PDE. For the case under consideration we may write:

```
[ > pdetest(Sol1,PDE1);
```

$$0$$

A zero value indicates that the expression contained in *Sol1* satisfies the PDE described by *PDE1*.

A second example of solving PDEs with *pdsolve* is the following:

```
[ > PDE2:=diff(f(x,y),x$2) = x + y;
```

$$PDE2 := \frac{\partial^2}{\partial x^2} f(x, y) = x + y$$

```
> Sol2:=pdsolve(PDE2,f(x,y));
```

$$Sol2 := f(x, y) = \frac{1}{6}x^3 + \frac{1}{2}y x^2 + _F1(y) x + _F2(y)$$

The type of solution provided for this example is referred to as a *quasi-general solution* for it involves a couple of arbitrary functions, namely, *_F1(y)* and *_F2(y)*, whose form cannot be determined from the PDE alone. Use of boundary conditions, for example, $f(0, y) = y$, and $f(1, y) = y^2$, may help in determining the expressions for

$_F1(y)$ and $_F2(y)$. Replacing the first boundary condition, $f(0, y) = y$ into the solution obtained above results in:

```
> Eq1:=subs({x=0,f(x,y)=y},Sol2);
```

$$Eq1 := y = _F2(y)$$

Thus, we can replace $_F2(y) = y$ in the solution to obtain:

```
> \_F2(y):=y;Sol2;
```

$$_F2(y) := y$$

$$f(x, y) = \frac{1}{6}x^3 + \frac{1}{2}yx^2 + _F1(y)x + y$$

The second boundary condition, $f(1, y) = y^2$, when replaced into the solution (*Sol2*), produces:

```
> Eq2:=subs({x=1,f(x,y)=y^2},Sol2);
```

$$Eq2 := y^2 = \frac{1}{6} + \frac{3}{2}y + _F1(y)$$

which results in:

```
> \_F1(y):=solve(Eq2,\_F1(y));
```


$$_F1(y) := y^2 - \frac{1}{6} - \frac{3}{2}y$$

The solution now being equal to:

```
> Sol2;
```

$$f(x, y) = \frac{1}{6}x^3 + \frac{1}{2}yx^2 + \left(y^2 - \frac{1}{6} - \frac{3}{2}y\right)x + y$$

A plot of this solution is shown below:



If *pdsolve* cannot find a general or quasi-general solution, it attempts a *separation of variables* of the PDE. For example,

```
> PDE3:=diff(u(x,y),x$2)=c^2*diff(u(x,y),y);
```

$$PDE3 := \frac{\partial^2}{\partial x^2} u(x, y) = c^2 \left(\frac{\partial}{\partial y} u(x, y) \right)$$

```
> Sol3:=pdsolve(PDE3,u(x,y));
```

$$\text{Sol3} := (u(x, y) = _F3(x) _F4(y)) \&\text{where} \left[\left\{ \frac{\partial}{\partial y} _F4(y) = \frac{-c_1 _F4(y)}{c^2}, \frac{\partial^2}{\partial x^2} _F3(x) = -c_1 _F3(x) \right\} \right]$$

The result shows that the solution can be written as the product of two functions $_F1(x)$ and $_F2(y)$. The ordinary differential equations (ODE) that the functions $_F1$ and $_F2$ satisfy are presented as a list after the particle *&where*.

If we want *pdsolve* to integrate the resulting ODEs we can use the option *INTEGRATE* as in this example:

```
> Sol3a:=pdsolve(PDE3,u(x,y),I NTEGRATE);
```

$$\text{Sol3a} := (u(x, y) = _F3(x) _F4(y)) \&\text{where} \left[\left\{ \left\{ _F3(x) = _C1 \sinh(\sqrt{-c_1} x) + _C2 \cosh(\sqrt{-c_1} x) \right\}, \left\{ _F4(y) = _C3 e^{\left(\frac{-c_1 y}{c^2}\right)} \right\} \right\} \right]$$

General solutions for the resulting ODEs are now reported in the output in place of the ODEs themselves. From this point on the user can use boundary conditions to determine the values of the constants involved in the functions $_F3$ and $_F4$. The constants $_c_1$ typically produce *eigenvalues* of the solution.

The function *pdsolve* permits the use of the option *HINT* to suggest possible forms of the solution. For example, the option *HINT = `+`* suggest a separation-of-variable solution in which the solution to the PDE, say $u(x, y)$, is of the form $F(x) + G(y)$. The following example shows the use of the option *HINT = `+`*. Note: the apostrophe used around the plus sign is the back apostrophe (```):

```
> PD4:=diff(u(x,y),x$2)+diff(u(x,y),x)+diff(u(x,y),y) = 0;
```

$$\text{PD4} := \left(\frac{\partial^2}{\partial x^2} u(x, y) \right) + \left(\frac{\partial}{\partial x} u(x, y) \right) + \left(\frac{\partial}{\partial y} u(x, y) \right) = 0$$

```
> Sol4:=pdsolve(PD4,u(x,y),HI NT=`+`);
```

$$\text{Sol4} := (u(x, y) = _F3(x) + _F4(y)) \&\text{where} \left[\left\{ \frac{\partial}{\partial y} _F4(y) = -c_2, \frac{\partial^2}{\partial x^2} _F3(x) = -\left(\frac{\partial}{\partial x} _F3(x) \right) - c_2 \right\} \right]$$

If the integrated functions $_F3$ and $_F4$ are desired use:

```
> Sol4a:=pdsolve(PD4,u(x,y),HI NT=`+`,I NTEGRATE);
```

$$\text{Sol4a} := (u(x, y) = _F3(x) + _F4(y)) \&\text{where} \left[\left\{ \left\{ _F3(x) = -c_2 x + _C1 + _C2 e^{(-x)} \right\}, \left\{ _F4(y) = -c_2 y + _C3 \right\} \right\} \right]$$

Without using the *HINT* option, the result is:

```
> Sol4b:=pdsolve(PD4,u(x,y));
```

$$\text{Sol4b} := (u(x, y) = _F3(x) _F4(y)) \&\text{where} \left[\left\{ \frac{\partial}{\partial y} _F4(y) = -c_1 _F4(y), \frac{\partial^2}{\partial x^2} _F3(x) = -c_1 _F3(x) - \left(\frac{\partial}{\partial x} _F3(x) \right) \right\} \right]$$

This same result could be obtained by using the option *HINT = `*`*:

```
> Sol4c:=pdsolve(PD4,u(x,y),HI NT=`*`);
```

$$\text{Sol4c} := (u(x, y) = _F3(x) _F4(y)) \&\text{where} \left[\left\{ \frac{\partial}{\partial y} _F4(y) = -c_1 _F4(y), \frac{\partial^2}{\partial x^2} _F3(x) = -c_1 _F3(x) - \left(\frac{\partial}{\partial x} _F3(x) \right) \right\} \right]$$

If you prefer to provide your own names for the functions resulting from a separation of variables, e.g., $u(x, y) = X(x) Y(y)$, use:

```
> Sol4d:=pdsolve(PD4,u(x,y),HI NT=X(x)*Y(y));
```

$$\left[\text{Sol4d} := (u(x, y) = X(x) Y(y)) \&\text{where} \left[\left\{ \frac{\partial^2}{\partial x^2} X(x) = -c_1 X(x) - \left(\frac{\partial}{\partial x} X(x) \right), \frac{\partial}{\partial y} Y(y) = -Y(y) - c_1 \right\} \right] \right]$$

You can use the *HINT* option to provide other hints about the functions in the solution, for example,

$$\left[\begin{aligned} > \text{Sol4e} := \text{pdsolve}(\text{PD4}, u(x, y), \text{HINT} = \exp(k*x)*f(y)); \\ \text{Sol4e} := (u(x, y) = e^{(k*x)} f(y)) \&\text{where} \left[\left\{ \frac{\partial}{\partial y} f(y) = -k^2 f(y) - k f(y) \right\} \right] \end{aligned} \right]$$

Another possibility for *HINT* is:

$$\left[\begin{aligned} > \text{Sol4f} := \text{pdsolve}(\text{PD4}, u(x, y), \text{HINT} = \exp(k*x)*\exp(r*y)); \\ \text{Sol4f} := \end{aligned} \right]$$

However, *pdsolve* could not find a solution for functions of the form suggested (i.e., $u(x, y) = \exp(k*x)*\exp(r*y)$), therefore, it returns an empty value.

To find out more about the *PDEtools* package and the functions it includes use:

$$\left[> ?\text{PDEtools} \right]$$

Exercises

In the following exercises we use the following PDEs from physical science:

One-dimensional wave equation:	$u_{tt} = c^2 u_{xx}$
One-dimensional heat equation:	$u_t = c^2 u_{xx}$
Two-dimensional Laplace equation:	$u_{xx} + u_{yy} = 0$
Two-dimensional Poisson equation:	$u_{xx} + u_{yy} = f(x, y)$

With the notation $u_t = \partial u / \partial t$, $u_{xx} = \partial^2 u / \partial x^2$, etc.

[1]. Determine if the following functions $u(x, t)$ satisfy the one-dimensional wave equation:

(a) $u(x, t) = cxt$	(b) $u(x, t) = \ln(x^2 - c^2 t^2)$
(c) $u(x, t) = x^2 + c^2 t^2$	(d) $u(x, t) = x / (x^2 + c^2 t^2)$

[2]. Determine if the following functions $u(x, t)$ satisfy the one-dimensional heat equation:

(a) $u(x, t) = \exp(k(x + kc^2 t))$	(b) $u(x, t) = t + x^2 / (2c^2)$
(c) $u(x, t) = 2c^2 / (2c^2 t + x^2)$	(d) $u(x, t) = x^2 + 2c^2 t^2$

[3]. Determine if the following functions $u(x, y)$ satisfy the two-dimensional Laplace equation:

(a) $u(x, y) = k(x^2 + y^2)$	(b) $u(x, y) = \tan^{-1}(y/x)$
(c) $u(x, y) = \ln(x^2 + y^2)$	(d) $u(x, y) = x^3 - 3xy^2$

[3]. Determine if the functions $u(x, y)$ satisfy the two-dimensional Poisson equation for the given function $f(x, y)$:

(a) $u(x, y) = x \cos y$,	(a) $f(x, y) = -x \cos y$
(b) $u(x, y) = x^2 - 2xy + y^2$,	(b) $f(x, y) = 4$
(c) $u(x, y) = x \ln y$,	(c) $f(x, y) = -x/y^2$
(d) $u(x, y) = (x+y)^3$,	(d) $f(x, y) = 12(x+y)$

[4]. Classify the following second-order partial differential equations as parabolic, hyperbolic, or elliptic:

(a) $u_t - u_{tx} - u_{xx} = 0$	(b) $u_{xx} - 2u_{xy} + u_{yy} = xy$
(c) $u_{xx} + u_{xt} - u_{tt} + u_x + u_t = 0$	(d) $u_{xx} + u_{yy} = 0$

[5]. Let $S(\omega)$ and $C(\omega)$ be the coefficients of the sine and cosine Fourier transform, respectively, for a non-periodic function. Determine the coefficients $C(\omega)$, $S(\omega)$, and the continuous spectrum $A(\omega)$, for the following functions $f(x)$ for $x > 0$, and $f(x) = 0$, $x < 0$:

(a) $f(x) = \exp(-x^2/2)$	(b) $f(x) = \ln(x+1)$	(c) $f(x) = 1/(1+x^2)$	(d) $f(x) = \tan(x)$
---------------------------	-----------------------	------------------------	----------------------

[6]. Determine the Fourier transform of the function $f(t)$ for $t > 0$, and $f(t) = 0$, for $t < 0$, plot the real and imaginary components of the transform:

(a) $f(t) = t/(t^2+1)$	(b) $f(t) = \exp(t^2)$	(c) $f(t) = (t+1)/(t+2)$	(e) $f(t) = (1+t^2)^{1/2}$
------------------------	------------------------	--------------------------	----------------------------

[7]. **Solution to the heat equation with derivative boundary conditions.** Using separation of variables, solve the one-dimensional heat equation for the temperature $u(x, t)$ on a bar of length $L = 1$, using as boundary conditions $\partial u / \partial x = 0$ for $x = 0$ and $x = L$. Assume that the initial temperature distribution is given by $u(x, 0) = f(x) = x(1-x)$, and use $k=1$.

[8]. **Solution to the heat equation with heat convection loss.** Consider the one-dimensional heat equation including convection losses: $\partial u / \partial t = k \cdot (\partial^2 u / \partial x^2) - \beta u$, and subject to the boundary conditions: $u(0, t) = 0$, $u(L, t) = 0$, and the initial condition: $u(x, 0) = f(x)$.

- (a) Using the transformation $u(x,t) = \exp(-\beta \cdot t) \cdot w(x,t)$, verify that the original PDE is transformed into $\partial w / \partial t = k \cdot (\partial^2 w / \partial x^2)$, and that the corresponding boundary and initial conditions transform into $w(0,t) = 0$, $w(L,t) = 0$, and $w(x,0) = f(x)$.
- (b). For $L = 1$, $k = 1$, $\beta = 0.01$, and $f(x) = x(1-x)$ determine the solution $u(x,t)$ to the heat equation with convection loss.

[9]. **Solution to the diffusion-convection equation with constant flow velocity.** Many equations related to diffusion in moving fluids include a term involving the flow velocity v in the x -direction: $\partial u / \partial t + v \cdot (\partial u / \partial x) = k \cdot (\partial^2 u / \partial x^2)$. If v is a constant, this equation can be reduced to the form: $\partial w / \partial t = k \cdot (\partial^2 w / \partial x^2)$, by using the transformation


$$u(x,t) = \exp\left[\frac{v}{2 \cdot k} \cdot \left(x - \frac{1}{2} \cdot v \cdot t\right)\right] \cdot w(x,t).$$

- (a) Verify that the transformation indicated above produces the modified PDE: $\partial w / \partial t = k \cdot (\partial^2 w / \partial x^2)$.
- (b) Solve the diffusion-convection equation with constant flow velocity in the domain $0 < x < L$ for $L = 1$, $k = 1$, $v = 1$, and $k = 1$, subject to the boundary conditions $u(0,t) = 0$, and $u(L,t) = 0$, and initial conditions $u(x,0) = x(1-x)$.

[10]. Solve the one-dimensional wave equation for $c = 1$ in the domain $0 < x < L$, with $L = 1$, given the boundary conditions $u(0,t) = 0$, and $\partial u / \partial x = 0$ at $x = L$, and the initial condition $u(x,0) = f(x) = \cos(x)$.

[11]. Prove that $u(x,t) = F(x+ct) + F(x-ct)$ is a solution to the one-dimensional wave equation. Plot the result if $F(x) = \exp(-x^2/2)$ with $c = 1$ in the interval $-10 < x < 10$, and $-10 < t < 10$. Produce a three-dimensional surface plot as well as an animation of the results.

[12]. Referring to the following figure, solve case [3] of Laplace's equation in the rectangular domain $L = 2$, $H = 1$, if $p(y) = (y/H)(1-y/H)$. Produce a surface plot as well as a contour plot of the results.



[13]. Referring to the figure of problem [12], solve case [4] of Laplace's equation in the rectangular domain $L = 2$, $H = 1$, if $q(y) = (y/H)^2(1-y/H)$. Produce a surface plot as well as a contour plot of the results.

[14]. Using the results of problems [12] and [14] determine the solution to Laplace's equation in the rectangular domain $L = 2$, $H = 1$, if the boundary conditions are $u(x,0) = u(x,H) = 0$, $u(0,y) = p(y) = (y/H)(1-y/H)$, and $u(L,y) = q(y) = (y/H)^2(1-y/H)$. Produce a surface plot as well as a contour plot of the results.

[15]. Determine the sine, cosine, and Fourier transform for the following functions:

(a) $f(t) = t^2$ (b) $f(t) = \exp(-|t|)$ (c) $f(t) = \sin(t)/t$ (d) $f(t) = 1/(1+t^2)$

[16]. Use Fourier transforms to obtain the convolution of the following functions:

(a) $f(t) = t^2$, $g(t) = \exp(-t^2)$ (b) $f(t) = \sin(t)$, $g(t) = \exp(-t^2/2)$
(c) $f(t) = t^2$, $g(t) = \exp(-|t|)$ (d) $f(t) = \sin(t)$, $g(t) = 1+t^2$

REFERENCES

- Farlow, Stanley J., 1982, "*Partial Differential Equations for Scientists and Engineers*," Dover Publications Inc., New York.
- Friedman, B., 1956 (reissued 1990), "*Principles and Techniques of Applied Mathematics*," Dover Publications Inc., New York.
- Gullberg, J., 1997, "*Mathematics - From the Birth of Numbers*," W. W. Norton & Company, New York.
- Harris, J.W., and H. Stocker, 1998, "*Handbook of Mathematics and Computational Science*," Springer, New York.
- Heal, K.M., M.L. Hansen, and K. M. Rickard, 1998, "*Maple V - Learning Guide*," Springer, New York.
- Kamerich, E., 1999, "*A Guide to Maple*," Springer, New York.
- Keener, J. P., 1998, "*Principles of Applied Mathematics*," Perseus Books, Reading, Massachusetts.
- Kottegoda, N. T., and R. Rosso, 1997, "*Probability, Statistics, and Reliability for Civil and Environmental Engineers*," The Mc-Graw Hill Companies, Inc., New York.
- Kreysig, E., 1983, "*Advanced Engineering Mathematics - Fifth Edition*," John Wiley & Sons, New York.
- Monangan, M.B., K.O. Geddes, K.M. Heal, G. Labahn, and S.M. Vorkoetter, 1998, "*Maple V - Programming Guide*," Springer, New York.
- Newland, D.E., 1993, "*An Introduction to Random Vibrations, Spectral & Wavelet Analysis - Third Edition*," Longman Scientific and Technical, New York.
- Robertson J.S., 1996, "*Engineering Mathematics with Maple*," Mc Graw Hill, Inc., New York.
- Schwartz, D.I., 1999, "*Introduction to Maple*", E-Source, Prentice Hall, Upper Saddle River, New Jersey.
- Spiegel, M. R., 1971 (second printing, 1999), "*Schaum's Outline of Theory and Problems of Advanced Mathematics for Engineers and Scientists*," Schaum's Outline Series, McGraw-Hill, New York.
- Tinker, M. and R. Lambourne, 2000, "*Further Mathematics for the Physical Sciences*," John Wiley & Sons, LTD., Chichester, U.K.
- Tveito, A. and R. Winther, 1998, "*Introduction to Partial Differential Equations - A Computational Approach*," Texts in Applied Mathematics 29, Springer, New York.
- Urroz, G., 2000, "*Science and Engineering Mathematics with the HP 49 G - Volumes I & II*", www.greatunpublished.com, Charleston, S.C.